If your problem requires a series of related binomial coefficients, a good idea is to use recurrence relations, for example

$$\begin{aligned}\binom{n+1}{k} &= \frac{n+1}{n-k+1}\binom{n}{k} = \binom{n}{k} + \binom{n}{k-1}\\ \binom{n}{k+1} &= \frac{n-k}{k+1}\binom{n}{k}\end{aligned} \tag{6.1.7}$$

Finally, turning away from the combinatorial functions with integer valued arguments, we come to the beta function,

$$B(z,w) = B(w,z) = \int_0^1 t^{z-1}(1-t)^{w-1}dt \tag{6.1.8}$$

which is related to the gamma function by

$$B(z,w) = \frac{\Gamma(z)\Gamma(w)}{\Gamma(z+w)} \tag{6.1.9}$$

hence

```
      FUNCTION beta(z,w)
      REAL beta,w,z
C     USES gammln
        Returns the value of the beta function B(z,w).
      REAL gammln
      beta=exp(gammln(z)+gammln(w)-gammln(z+w))
      return
      END
```

CITED REFERENCES AND FURTHER READING:

Abramowitz, M., and Stegun, I.A. 1964, *Handbook of Mathematical Functions*, Applied Mathematics Series, Volume 55 (Washington: National Bureau of Standards; reprinted 1968 by Dover Publications, New York), Chapter 6.

Lanczos, C. 1964, *SIAM Journal on Numerical Analysis*, ser. B, vol. 1, pp. 86–96. [1]

6.2 Incomplete Gamma Function, Error Function, Chi-Square Probability Function, Cumulative Poisson Function

The incomplete gamma function is defined by

$$P(a,x) \equiv \frac{\gamma(a,x)}{\Gamma(a)} \equiv \frac{1}{\Gamma(a)}\int_0^x e^{-t}t^{a-1}dt \qquad (a>0) \tag{6.2.1}$$



Figure 6.2.1. The incomplete gamma function $P(a, x)$ for four values of a.

It has the limiting values

$$P(a, 0) = 0 \qquad \text{and} \qquad P(a, \infty) = 1 \tag{6.2.2}$$

The incomplete gamma function $P(a, x)$ is monotonic and (for a greater than one or so) rises from "near-zero" to "near-unity" in a range of x centered on about $a - 1$, and of width about $\sqrt{a}$ (see Figure 6.2.1).

The complement of $P(a, x)$ is also confusingly called an incomplete gamma function,

$$Q(a, x) \equiv 1 - P(a, x) \equiv \frac{\Gamma(a, x)}{\Gamma(a)} \equiv \frac{1}{\Gamma(a)} \int_x^\infty e^{-t} t^{a-1} dt \qquad (a > 0) \tag{6.2.3}$$

It has the limiting values

$$Q(a, 0) = 1 \qquad \text{and} \qquad Q(a, \infty) = 0 \tag{6.2.4}$$

The notations $P(a, x), \gamma(a, x)$, and $\Gamma(a, x)$ are standard; the notation $Q(a, x)$ is specific to this book.

There is a series development for $\gamma(a, x)$ as follows:

$$\gamma(a, x) = e^{-x} x^a \sum_{n=0}^{\infty} \frac{\Gamma(a)}{\Gamma(a + 1 + n)} x^n \tag{6.2.5}$$

One does not actually need to compute a new $\Gamma(a + 1 + n)$ for each n; one rather uses equation (6.1.3) and the previous coefficient.

A continued fraction development for $\Gamma(a,x)$ is

$$\Gamma(a,x) = e^{-x}x^a\left(\frac{1}{x+}\;\frac{1-a}{1+}\;\frac{1}{x+}\;\frac{2-a}{1+}\;\frac{2}{x+}\cdots\right) \qquad (x>0) \qquad (6.2.6)$$

It is computationally better to use the even part of (6.2.6), which converges twice as fast (see §5.2):

$$\Gamma(a,x) = e^{-x}x^a\left(\frac{1}{x+1-a-}\;\frac{1\cdot(1-a)}{x+3-a-}\;\frac{2\cdot(2-a)}{x+5-a-}\;\cdots\right) \qquad (x>0) \qquad (6.2.7)$$

It turns out that (6.2.5) converges rapidly for x less than about $a+1$, while (6.2.6) or (6.2.7) converges rapidly for x greater than about $a+1$. In these respective regimes each requires at most a few times $\sqrt{a}$ terms to converge, and this many only near $x=a$, where the incomplete gamma functions are varying most rapidly. Thus (6.2.5) and (6.2.7) together allow evaluation of the function for all positive a and x. An extra dividend is that we never need compute a function value near zero by subtracting two nearly equal numbers. The higher-level functions that return $P(a,x)$ and $Q(a,x)$ are

```
      FUNCTION gammp(a,x)
      REAL a,gammp,x
C     USES gcf,gser
         Returns the incomplete gamma function P(a,x).
      REAL gammcf,gamser,gln
      if(x.lt.0..or.a.le.0.)pause 'bad arguments in gammp'
      if(x.lt.a+1.)then             Use the series representation.
        call gser(gamser,a,x,gln)
        gammp=gamser
      else                          Use the continued fraction representation
        call gcf(gammcf,a,x,gln)
        gammp=1.-gammcf             and take its complement.
      endif
      return
      END
```

```
      FUNCTION gammq(a,x)
      REAL a,gammq,x
C     USES gcf,gser
         Returns the incomplete gamma function Q(a,x) ≡ 1 − P(a,x).
      REAL gammcf,gamser,gln
      if(x.lt.0..or.a.le.0.)pause 'bad arguments in gammq'
      if(x.lt.a+1.)then             Use the series representation
        call gser(gamser,a,x,gln)
        gammq=1.-gamser             and take its complement.
      else                          Use the continued fraction representation.
        call gcf(gammcf,a,x,gln)
        gammq=gammcf
      endif
      return
      END
```

The argument gln is returned by both the series and continued fraction procedures containing the value $\ln\Gamma(a)$; the reason for this is so that it is available to you if you want to modify the above two procedures to give $\gamma(a,x)$ and $\Gamma(a,x)$, in addition to $P(a,x)$ and $Q(a,x)$ (cf. equations 6.2.1 and 6.2.3).

The procedures gser and gcf which implement (6.2.5) and (6.2.7) are

```
      SUBROUTINE gser(gamser,a,x,gln)
      INTEGER ITMAX
      REAL a,gamser,gln,x,EPS
      PARAMETER (ITMAX=100,EPS=3.e-7)
C     USES gammln
         Returns the incomplete gamma function P(a,x) evaluated by its series representation as
         gamser. Also returns ln Γ(a) as gln.
      INTEGER n
      REAL ap,del,sum,gammln
      gln=gammln(a)
      if(x.le.0.)then
        if(x.lt.0.)pause 'x < 0 in gser'
        gamser=0.
        return
      endif
      ap=a
      sum=1./a
      del=sum
      do 11 n=1,ITMAX
        ap=ap+1.
        del=del*x/ap
        sum=sum+del
        if(abs(del).lt.abs(sum)*EPS)goto 1
      enddo 11
      pause 'a too large, ITMAX too small in gser'
1     gamser=sum*exp(-x+a*log(x)-gln)
      return
      END
```

```
      SUBROUTINE gcf(gammcf,a,x,gln)
      INTEGER ITMAX
      REAL a,gammcf,gln,x,EPS,FPMIN
      PARAMETER (ITMAX=100,EPS=3.e-7,FPMIN=1.e-30)
C     USES gammln
         Returns the incomplete gamma function Q(a,x) evaluated by its continued fraction repre-
         sentation as gammcf. Also returns ln Γ(a) as gln.
         Parameters: ITMAX is the maximum allowed number of iterations; EPS is the relative accu-
         racy; FPMIN is a number near the smallest representable floating-point number.
      INTEGER i
      REAL an,b,c,d,del,h,gammln
      gln=gammln(a)
      b=x+1.-a                      Set up for evaluating continued fraction by modified
      c=1./FPMIN                       Lentz's method (§5.2) with b0 = 0.
      d=1./b
      h=d
      do 11 i=1,ITMAX               Iterate to convergence.
        an=-i*(i-a)
        b=b+2.
        d=an*d+b
        if(abs(d).lt.FPMIN)d=FPMIN
        c=b+an/c
        if(abs(c).lt.FPMIN)c=FPMIN
        d=1./d
        del=d*c
```

```
      h=h*del
      if(abs(del-1.).lt.EPS)goto 1
   enddo 11
   pause 'a too large, ITMAX too small in gcf'
1  gammcf=exp(-x+a*log(x)-gln)*h      Put factors in front.
   return
   END
```

Error Function

The error function and complementary error function are special cases of the incomplete gamma function, and are obtained moderately efficiently by the above procedures. Their definitions are

$$\text{erf}(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2} dt \tag{6.2.8}$$

and

$$\text{erfc}(x) \equiv 1 - \text{erf}(x) = \frac{2}{\sqrt{\pi}} \int_x^\infty e^{-t^2} dt \tag{6.2.9}$$

The functions have the following limiting values and symmetries:

$$\text{erf}(0) = 0 \qquad \text{erf}(\infty) = 1 \qquad \text{erf}(-x) = -\text{erf}(x) \tag{6.2.10}$$

$$\text{erfc}(0) = 1 \qquad \text{erfc}(\infty) = 0 \qquad \text{erfc}(-x) = 2 - \text{erfc}(x) \tag{6.2.11}$$

They are related to the incomplete gamma functions by

$$\text{erf}(x) = P\left(\frac{1}{2}, x^2\right) \qquad (x \geq 0) \tag{6.2.12}$$

and

$$\text{erfc}(x) = Q\left(\frac{1}{2}, x^2\right) \qquad (x \geq 0) \tag{6.2.13}$$

Hence we have

```
   FUNCTION erf(x)
   REAL erf,x
C  USES gammp
      Returns the error function erf(x).
   REAL gammp
   if(x.lt.0.)then
      erf=-gammp(.5,x**2)
   else
      erf=gammp(.5,x**2)
   endif
   return
   END
```

```
      FUNCTION erfc(x)
      REAL erfc,x
C     USES gammp,gammq
        Returns the complementary error function erfc(x).
      REAL gammp,gammq
      if(x.lt.0.)then
          erfc=1.+gammp(.5,x**2)
      else
          erfc=gammq(.5,x**2)
      endif
      return
      END
```

If you care to do so, you can easily remedy the minor inefficiency in `erf` and `erfc`, namely that $\Gamma(0.5) = \sqrt{\pi}$ is computed unnecessarily when `gammp` or `gammq` is called. Before you do that, however, you might wish to consider the following routine, based on Chebyshev fitting to an inspired guess as to the functional form:

```
      FUNCTION erfcc(x)
      REAL erfcc,x
        Returns the complementary error function erfc(x) with fractional error everywhere less than
        1.2 × 10^-7.
      REAL t,z
      z=abs(x)
      t=1./(1.+0.5*z)
      erfcc=t*exp(-z*z-1.26551223+t*(1.00002368+t*(.37409196+
     *      t*(.09678418+t*(-.18628806+t*(.27886807+t*(-1.13520398+
     *      t*(1.48851587+t*(-.82215223+t*.17087277)))))))))
      if (x.lt.0.) erfcc=2.-erfcc
      return
      END
```

There are also some functions of *two* variables that are special cases of the incomplete gamma function:

Cumulative Poisson Probability Function

$P_x(< k)$, for positive x and integer $k \geq 1$, denotes the *cumulative Poisson probability* function. It is defined as the probability that the number of Poisson random events occurring will be between 0 and $k-1$ *inclusive*, if the expected mean number is x. It has the limiting values

$$P_x(< 1) = e^{-x} \qquad P_x(< \infty) = 1 \tag{6.2.14}$$

Its relation to the incomplete gamma function is simply

$$P_x(< k) = Q(k, x) = \texttt{gammq}\,(k, x) \tag{6.2.15}$$

Chi-Square Probability Function

$P(\chi^2|\nu)$ is defined as the probability that the observed chi-square for a correct model should be less than a value χ^2. (We will discuss the use of this function in Chapter 15.) Its complement $Q(\chi^2|\nu)$ is the probability that the observed chi-square will exceed the value χ^2 by chance *even* for a correct model. In both cases ν is an integer, the number of degrees of freedom. The functions have the limiting values

$$P(0|\nu) = 0 \qquad P(\infty|\nu) = 1 \tag{6.2.16}$$

$$Q(0|\nu) = 1 \qquad Q(\infty|\nu) = 0 \tag{6.2.17}$$

and the following relation to the incomplete gamma functions,

$$P(\chi^2|\nu) = P\left(\frac{\nu}{2}, \frac{\chi^2}{2}\right) = \texttt{gammp}\left(\frac{\nu}{2}, \frac{\chi^2}{2}\right) \tag{6.2.18}$$

$$Q(\chi^2|\nu) = Q\left(\frac{\nu}{2}, \frac{\chi^2}{2}\right) = \texttt{gammq}\left(\frac{\nu}{2}, \frac{\chi^2}{2}\right) \tag{6.2.19}$$

CITED REFERENCES AND FURTHER READING:

Abramowitz, M., and Stegun, I.A. 1964, *Handbook of Mathematical Functions*, Applied Mathematics Series, Volume 55 (Washington: National Bureau of Standards; reprinted 1968 by Dover Publications, New York), Chapters 6, 7, and 26.

Pearson, K. (ed.) 1951, *Tables of the Incomplete Gamma Function* (Cambridge: Cambridge University Press).

6.3 Exponential Integrals

The standard definition of the exponential integral is

$$E_n(x) = \int_1^\infty \frac{e^{-xt}}{t^n} dt, \qquad x > 0, \quad n = 0, 1, \ldots \tag{6.3.1}$$

The function defined by the principal value of the integral

$$\text{Ei}(x) = -\int_{-x}^\infty \frac{e^{-t}}{t}\, dt = \int_{-\infty}^x \frac{e^t}{t}\, dt, \qquad x > 0 \tag{6.3.2}$$

is also called an exponential integral. Note that $\text{Ei}(-x)$ is related to $-E_1(x)$ by analytic continuation.

The function $E_n(x)$ is a special case of the incomplete gamma function

$$E_n(x) = x^{n-1}\Gamma(1-n, x) \tag{6.3.3}$$